Analytical solution to the aeroelastic response of a two-dimensional elastic plate in axial potential flow

Online supplementary material

Mathematica script

Cory Medina[1,2] and Chang-kwon Kang[1]
1. Department of Mechanical and Aerospace Engineering, University of Alabama in Huntsville, Huntsville, AL 35899, USA
2. NASA Marshall Space Flight Center, Huntsville, AL 35812, USA

Airfoil deformation

M is the order of the Chebychev expansion, which can be set to any arbitrary positive integer. M=4 is shown in this example. Mathematica 11 is used.

In[1]:=
```
M = 4;
Do[Tn[n] = ChebyshevT[n, x*], {n, 0, M}];
ψi = 0; Do[ψi = ψi + ci[n] * Tn[n], {n, 0, M}];
w* = ψi * τi[t*];
v* = w* + h*[t*];
Framed[Row[{"v*(x*,t*) = ", v*}]]
```

Out[6]=

$$v^*(x^*,t^*) = \left(ci[0] + ci[1]\,x^* + ci[2]\left(-1 + 2\,(x^*)^2\right) + ci[3]\left(-3\,x^* + 4\,(x^*)^3\right) + ci[4]\left(1 - 8\,(x^*)^2 + 8\,(x^*)^4\right)\right)\tau i[t^*] + h^*[t^*]$$

Noncirculatory flow

Source/sink strength per unit length

In[7]:=
```
σ = 2 (U D[v*, x*] + b f D[v*, t*]);
Framed[Row[{"σ(x*,t*) = ", σ // Simplify}]]
```

Out[8]=

$$\sigma(x^*,t^*) = 2\left(U\left(ci[1] - 3\,ci[3] + 4\left(ci[2] - 4\,ci[4]\right)x^* + 12\,ci[3]\,(x^*)^2 + 32\,ci[4]\,(x^*)^3\right)\tau i[t^*] + b\,f\left(\left(ci[0] - ci[2] + ci[4] + \left(ci[1] - 3\,ci[3]\right)x^* + 2\left(ci[2] - 4\,ci[4]\right)(x^*)^2 + 4\,ci[3]\,(x^*)^3 + 8\,ci[4]\,(x^*)^4\right)\tau i'[t^*] + (h^*)'[t^*]\right)\right)$$

Velocity potential for source/sink pair

In[9]:=

$$(*\ \phi nc(x^*,t^*) = \frac{b}{4\pi}\int_{-1}^{1}\sigma(x_1^*,t^*) * \mathrm{Log}\left[\frac{(x^*-x_1^*)^2 + \left(\sqrt{1-x^{*2}} - \sqrt{1-x_1^{*2}}\right)^2}{(x^*-x_1^*)^2 + \left(\sqrt{1-x^{*2}} + \sqrt{1-x_1^{*2}}\right)^2}\right] dx_1^*\ *)$$

It is easier for Mathematica to evaluate the integral term by term. We tabulate the first M+1 terms.

In[10]:=

```
For[n = 0, n ≤ M, n++, Int[n] = Integrate[x1*^n Log[((x* - x1*)^2 + (Sqrt[1 - x*^2] - Sqrt[1 - x1*^2])^2)/((x* - x1*)^2 + (Sqrt[1 - x*^2] + Sqrt[1 - x1*^2])^2)],
    {x1*, -1, 1}, Assumptions → {-1 <= x* <= 1}, PrincipalValue → True] // Factor];
For[n = 0, n ≤ M, n++, Print[Int[n]]]
```

$$-2\pi\sqrt{1-(x^*)^2}$$

$$-\pi x^*\sqrt{1-(x^*)^2}$$

$$\frac{\pi(-1+x^*)(1+x^*)\left(1+2(x^*)^2\right)}{3\sqrt{1-(x^*)^2}}$$

$$\frac{\pi(-1+x^*)x^*(1+x^*)\left(1+2(x^*)^2\right)}{4\sqrt{1-(x^*)^2}}$$

$$\frac{\pi(-1+x^*)(1+x^*)\left(3+4(x^*)^2+8(x^*)^4\right)}{20\sqrt{1-(x^*)^2}}$$

Evaluate the noncirculatory potential.

In[12]:=

```
ϕnc = 0; Do[ϕnc = ϕnc + b/(4 π) Int[n] * Coefficient[σ, x*, n], {n, 0, M}]
Framed[Row[{"ϕnc(x*,t) = ", ϕnc // Simplify}]]
```

Out[13]=

$$\phi_{nc}(x^*,t) = -\frac{1}{30} b\sqrt{1-(x^*)^2}\Big(30U\left(ci[1]-ci[3]+2\left(ci[2]-2ci[4]\right)x^*+4ci[3](x^*)^2+8ci[4](x^*)^3\right)\tau i[t^*]+ bf\Big(\left(30ci[0]-20ci[2]+8ci[4]+15\left(ci[1]-2ci[3]\right)x^*+ 4\left(5ci[2]-14ci[4]\right)(x^*)^2+30ci[3](x^*)^3+48ci[4](x^*)^4\right)\tau i'[t^*]+30(h^*)'[t^*]\Big)\Big)$$

Pressure difference for noncirculatory flow (Linearized Bernoulli's equation)

In[14]:=

```
Δpnc = 2 ρ (U/b D[ϕnc, x*] + f D[ϕnc, t*]);
Framed[Row[{"Δpnc(x,t) = ", Δpnc // Simplify}]]
```

Out[15]=

$$\Delta p_{nc}(x,t) = \frac{1}{15\sqrt{1-(x^*)^2}} \rho\Big(30U^2\left(-2ci[2]+4ci[4]+\left(ci[1]-9ci[3]\right)x^*+4\left(ci[2]-8ci[4]\right)(x^*)^2+ 12ci[3](x^*)^3+32ci[4](x^*)^4\right)\tau i[t^*]+ bf\Big(15U\left(-3ci[1]+4ci[3]+2\left(ci[0]-4ci[2]+8ci[4]\right)x^*+4\left(ci[1]-5ci[3]\right)(x^*)^2+ 8\left(ci[2]-6ci[4]\right)(x^*)^3+16ci[3](x^*)^4+32ci[4](x^*)^5\right)\tau i'[t^*]+15bf\left(ci[1]-4ci[3]\right) (x^*)^3\tau i''[t^*]+4bf\left(5ci[2]-26ci[4]\right)(x^*)^4\tau i''[t^*]+30bf\,ci[3](x^*)^5\tau i''[t^*]+ 48bf\,ci[4](x^*)^6\tau i''[t^*]+15x^*\left(2U(h^*)'[t^*]-bf\left(ci[1]-2ci[3]\right)\tau i''[t^*]\right)- 2bf\left(\left(15ci[0]-10ci[2]+4ci[4]\right)\tau i''[t^*]+15(h^*)''[t^*]\right)+ 2bf(x^*)^2\left(\left(15ci[0]-20ci[2]+32ci[4]\right)\tau i''[t^*]+15(h^*)''[t^*]\right)\Big)\Big)$$

Non circulatory lift

In[16]:=
```
Lnc = (int[b * Δpnc // Expand]
      //. int[a_ + b_] :> int[a] + int[b]
     //. int[i_] :> Integrate[i, {x*, -1, 1}, GenerateConditions → False]);
Framed[Row[{"Lnc'(t*) = ", Lnc // Simplify}]]
```

Out[17]=

$$L_{nc}'(t^*) = -\frac{1}{2} b^2 f \pi \rho \left(2\,U\,ci[1]\,\tau i'[t^*] + b\,f\left((2\,ci[0] - ci[2])\,\tau i''[t^*] + 2\,(h^*)''[t^*]\right)\right)$$

Circulatory flow

Kutta condition at TE

In[18]:=
```
(* ∂x*ϕc + ∂x*ϕnc = finite @ x*=1 *)
```

$$(*\ K = \frac{1}{2\pi}\int_1^\infty \sqrt{\frac{x_0^*+1}{x_0^*-1}}\,\gamma_w(x_0^*, t^*)\,dx_0^*\ *)$$

$$K = \text{Limit}\left[-D[\phi nc, x^*]\,\frac{\sqrt{1-x^{*2}}}{b}, x^* \to 1\right];$$

```
Framed[Row[{"K(t*) = ", K // Simplify}]]
```

Out[19]=

$$K(t^*) = -U\,(ci[1] + 2\,ci[2] + 3\,ci[3] + 4\,ci[4])\,\tau i[t^*] - \frac{1}{2} b\,f\left((2\,ci[0] + ci[1])\,\tau i'[t^*] + 2\,(h^*)'[t^*]\right)$$

Pressure difference for vortex pair in terms of K and C(k)

In[20]:=

$$(*\ \text{Theodorsen's Function: } C(k) = \frac{\int_1^\infty \left(\frac{x0}{\sqrt{x0^2-1}}\right)\gamma[x0,t]\,dx0}{\int_1^\infty \left(\sqrt{\frac{x0+1}{x0-1}}\right)\gamma[x0,t]\,dx0}\ *)$$

$$\Delta pc = 2\,\rho\,U\,K\,\frac{Ck + x^*\,(1 - Ck)}{\sqrt{1 - x^{*2}}};$$

```
Framed[Row[{"Δpc(x*,t*) = ", Δpc // Simplify}]]
```

Out[21]=

$$\Delta p_c(x^*, t^*) = \frac{1}{\sqrt{1-(x^*)^2}} U\rho\,(-Ck + (-1 + Ck)\,x^*)\left(2\,U\,(ci[1] + 2\,ci[2] + 3\,ci[3] + 4\,ci[4])\,\tau i[t^*] + b\,f\left((2\,ci[0] + ci[1])\,\tau i'[t^*] + 2\,(h^*)'[t^*]\right)\right)$$

Circulatory lift force

In[22]:=
```
Lc = (int[b * Δpc // Expand]
      //. int[a_ + b_] :> int[a] + int[b]
     //. int[i_] :> Integrate[i, {x*, -1, 1}, GenerateConditions → False]);
Framed[Row[{"Lc(t*)' = ", Lc // Simplify}]]
```

Out[23]=

$$L_c(t^*)' = -b\,Ck\,\pi\,U\,\rho\left(2\,U\,(ci[1] + 2\,ci[2] + 3\,ci[3] + 4\,ci[4])\,\tau i[t^*] + b\,f\left((2\,ci[0] + ci[1])\,\tau i'[t^*] + 2\,(h^*)'[t^*]\right)\right)$$

Total pressure difference

In[24]:=
```
Δp = Δpnc + Δpc;
Framed[Row[{"Δp(x*,t*) = ", Δp // Simplify}]]
```

Out[25]=

$$\Delta p(x^*,t^*) \;=\; \frac{1}{15\sqrt{1-(x^*)^2}} \rho\,(-1+x^*)\Big(30\,U^2\big(2\,(ci[2]-2\,ci[4]) + Ck\,(ci[1]+2\,ci[2]+3\,ci[3]+4\,ci[4]) + 4\,(ci[2]+3\,ci[3])\,x^* + 4\,(3\,ci[3]+8\,ci[4])\,(x^*)^2 + 32\,ci[4]\,(x^*)^3\big)\,\tau i[t^*] + b\,f\,\Big(15\,U\,\big(3\,ci[1] + Ck\,(2\,ci[0]+ci[1]) - 4\,ci[3] + 4\,(ci[1]+2\,ci[2]-ci[3]-4\,ci[4])\,x^* + 8\,(ci[2]+2\,ci[3]-2\,ci[4])\,(x^*)^2 + 16\,(ci[3]+2\,ci[4])\,(x^*)^3 + 32\,ci[4]\,(x^*)^4\big)\,\tau i'[t^*] + 30\,Ck\,U\,(h^*)'[t^*] + b\,f\,(1+x^*)\Big(\big(30\,ci[0]-20\,ci[2]+8\,ci[4]+15\,(ci[1]-2\,ci[3])\,x^* + 4\,(5\,ci[2]-14\,ci[4])\,(x^*)^2 + 30\,ci[3]\,(x^*)^3 + 48\,ci[4]\,(x^*)^4\big)\,\tau i''[t^*] + 30\,(h^*)''[t^*]\Big)\Big)\Big)$$

- **Verify Kutta condition: Δp(x^*=1) = finite**

In[26]:=
```
Framed[Row[{"Δp(x*->1) = ", Limit[Δp, x* → 1] // Simplify}]]
```

Out[26]=

$$\Delta p(x^*\text{->}1) \;=\; 0$$

Fluid force of of 2-way coupled aeroelastic model

In[27]:=
```
ψj = 0; Do[ψj = ψj + cj[n] * Tn[n], {n, 0, M}] ;
Δp* = Δp / (ρ U^2) ;
Qj* = (int[Δp* * ψj // Expand]
      //. int[a_ + b_] :→ int[a] + int[b]
     //. int[i_] :→ Integrate[i, {x*, -1, 1}, GenerateConditions → False]);
Framed[Row[{"Qj*(t*) = ", Qj* // Simplify}]]
```

Out[30]=

$$Q^*_j(t^*) \;=\; -\pi\,\Big(Ck\,(ci[1]+2\,ci[2]+3\,ci[3]+4\,ci[4])\,(2\,cj[0]-cj[1]) + 3\,ci[3]\,cj[1] + 4\,ci[4]\,cj[1] + 2\,ci[2]\,(cj[1]-cj[2]) - 3\,ci[3]\,cj[3] - 4\,ci[4]\,cj[4]\Big)\,\tau i[t^*] + \frac{1}{240\,U^2}\,b\,f\,\pi\,\Big(-120\,U\,\Big(Ck\,(2\,ci[0]+ci[1])\,(2\,cj[0]-cj[1]) + ci[1]\,(2\,cj[0]+cj[1]-2\,cj[2]) + 2\,\big(ci[2]\,(cj[1]-cj[3]) + ci[4]\,cj[3] + ci[3]\,(cj[2]-cj[4])\big)\Big)\,\tau i'[t^*] - 240\,Ck\,U\,(2\,cj[0]-cj[1])\,(h^*)'[t^*] + b\,f\,\Big(\big(-30\,ci[1]\,cj[1] + 30\,ci[3]\,cj[1] - 120\,ci[0]\,(2\,cj[0]-cj[2]) + 20\,ci[4]\,cj[2] + 30\,ci[1]\,cj[3] - 45\,ci[3]\,cj[3] - 32\,ci[4]\,cj[4] + 20\,ci[2]\,(6\,cj[0]-4\,cj[2]+cj[4])\big)\,\tau i''[t^*] + 120\,(-2\,cj[0]+cj[2])\,(h^*)''[t^*]\Big)\Big)$$

- **Total Lift force**

In[31]:=
```
L = Lnc + Lc // Simplify;
Framed[Row[{"L'(t*) = ", Collect[L, Ck, Simplify]}]]
```

Out[32]=

$$L'(t^*) \;=\; b\,Ck\,\pi\,U\,\rho\,\Big(-2\,U\,(ci[1]+2\,ci[2]+3\,ci[3]+4\,ci[4])\,\tau i[t^*] - b\,f\,\big((2\,ci[0]+ci[1])\,\tau i'[t^*] + 2\,(h^*)'[t^*]\big)\Big) - \frac{1}{2}\,b^2\,f\,\pi\,\rho\,\Big(2\,U\,ci[1]\,\tau i'[t^*] + b\,f\,\big((2\,ci[0]-ci[2])\,\tau i''[t^*] + 2\,(h^*)''[t^*]\big)\Big)$$

▪ Thrust

In[33]:=
```
dvdx = D[v*, x*];
Tp = (int[b * Δp * dvdx // Expand]
     //. int[a_ + b_] :→ int[a] + int[b]
    //. int[i_] :→ Integrate[i, {x*, -1, 1}, GenerateConditions → False]);
```

In[35]:=
```
S = √2/2 K (2 Ck - 1) + Limit[∂_{x*} ϕnc (1/b) √(1 + x*), x* → -1];
(*Leading-edge suction (von Karman & Burgers 1935)*)
TLES = π ρ b S^2;
```

In[37]:=
```
Framed[Row[{"T_p(t*) = ", Tp // Simplify}]]
Framed[Row[{"T_LES(t*) = ", TLES // Simplify}]]
```

Out[37]=
```
T_p(t*) = -b π ρ τi[t*] (2 U^2 (4 (ci[2] + 2 ci[4])^2 +
        Ck (ci[1]^2 - 4 ci[2]^2 + 6 ci[1] ci[3] + 9 ci[3]^2 - 16 ci[2] ci[4] - 16 ci[4]^2)) τi[t*] +
    b f (U (ci[1]^2 + 4 ci[2]^2 + 6 ci[3]^2 + Ck (2 ci[0] + ci[1]) (ci[1] - 2 ci[2] + 3 ci[3] - 4 ci[4]) +
          8 ci[4]^2 + ci[1] (2 ci[2] - 3 ci[3] + 4 ci[4])) τi′[t*] +
        2 Ck U (ci[1] - 2 ci[2] + 3 ci[3] - 4 ci[4]) (h*)′[t*] + b f ci[1] (ci[0] τi″[t*] + (h*)″[t*])))
```

Out[38]=
```
T_LES(t*) = 1/2 b π ρ (2 U (-2 (ci[2] + 2 ci[4]) + Ck (ci[1] + 2 ci[2] + 3 ci[3] + 4 ci[4])) τi[t*] +
      b f ((2 Ck ci[0] - ci[1] + Ck ci[1]) τi′[t*] + 2 Ck (h*)′[t*]))^2
```

▪ Power

In[39]:=
```
dvdt = f D[v*, t*];
P = (int[b^2 Δp * dvdt // Expand]
     //. int[a_ + b_] :→ int[a] + int[b]
    //. int[i_] :→ Integrate[i, {x*, -1, 1}, GenerateConditions → False]);
```

In[41]:=
```
Framed[Row[{"P'(t*) = ", P // Simplify}]]
```

Out[41]=
```
P'(t*) =
 -1/240 b^2 f π ρ (240 U^2 τi[t*] ((-2 ci[2]^2 - 3 ci[3]^2 - 4 ci[4]^2 + ci[1] (2 ci[2] + 3 ci[3] + 4 ci[4]) +
          Ck (2 ci[0] - ci[1]) (ci[1] + 2 ci[2] + 3 ci[3] + 4 ci[4])) τi′[t*] +
       2 Ck (ci[1] + 2 ci[2] + 3 ci[3] + 4 ci[4]) (h*)′[t*]) +
    b f (120 U (2 ci[0] + ci[1]) (2 Ck ci[0] + ci[1] - Ck ci[1]) τi′[t*]^2 +
       120 (h*)′[t*] (4 Ck U (h*)′[t*] + b f ((2 ci[0] - ci[2]) τi″[t*] + 2 (h*)″[t*])) +
       τi′[t*] (240 U (4 Ck ci[0] + ci[1]) (h*)′[t*] +
          b f ((240 ci[0]^2 + 30 ci[1]^2 - 240 ci[0] ci[2] + 80 ci[2]^2 - 60 ci[1] ci[3] + 45 ci[3]^2 -
                40 ci[2] ci[4] + 32 ci[4]^2) τi″[t*] + 120 (2 ci[0] - ci[2]) (h*)″[t*]))))
```

Summation symbol verifications

In[42]:=

```
Σ1i = Sum[i ci[i], {i, 1, M}];
Σ2i = Sum[(i ci[i]), {i, 2, M, 2}];
Σ3i = Sum[(i ci[i]), {i, 1, M, 2}];
Σ4i = Sum[(i/2 ci[i]), {i, 2, M, 2}];
Σ5ji = Sum[i ci[i] cj[i], {i, 1, M}];
Σ5ii = Sum[i ci[i]^2, {i, 1, M}];
Σ6ji = Sum[i/(2 (i - 1) (i + 1)) ci[i] cj[i], {i, 2, M}];
Σ6ii = Sum[i/(2 (i - 1) (i + 1)) ci[i]^2, {i, 2, M}];
Σ7ji = Sum[1/(4 (i + 1)) (ci[i + 2] cj[i] + ci[i] cj[i + 2]), {i, 1, M - 2}];
Σ7ii = Sum[1/(2 (i + 1)) (ci[i] ci[i + 2]), {i, 1, M - 2}];
Σ8ji = Sum[(ci[i + 1] cj[i] - ci[i] cj[i + 1]), {i, 1, M - 1}];
```

- **Lift summation verification**

In[53]:=

```
Lver = -π ρ b^2 (b f^2 (h*)''[t*] + 1/2 b (2 ci[0] - ci[2]) f^2 τi''[t*] + U ci[1] f τi'[t*]);
Lver = Lver - 2 π ρ U b Ck (b f (h*)'[t*] + 1/2 b (2 ci[0] + ci[1]) f τi'[t*] + U Σ1i τi[t*]);
L - Lver // Simplify (* should be zero *)
```

Out[55]= 0

- **Thrust summation verification**

In[56]:=

```
TLESver = 2 U^2 (Ck Σ1i - 2 Σ4i)^2 τi[t*]^2;
TLESver = TLESver + 2 b^2 Ck (Ck (2 ci[0] + ci[1]) - ci[1]) f^2 τi'[t*] (h*)'[t*];
TLESver = TLESver + 1/2 b^2 (ci[1] - Ck (2 ci[0] + ci[1]))^2 f^2 τi'[t*]^2;
TLESver = TLESver + 4 U b Ck (Ck Σ1i - 2 Σ4i) f τi[t*] (h*)'[t*];
TLESver = TLESver + 2 U b (Ck (2 ci[0] + ci[1]) - ci[1]) (Ck Σ1i - 2 Σ4i) f τi[t*] τi'[t*];
TLESver = TLESver + 2 b^2 Ck^2 f^2 (h*)'[t*]^2;
TLESver = (π ρ b) TLESver;
TLES - TLESver // Simplify (* Should be zero *)
```

Out[63]= 0

```
In[64]:= Tpver = 2 U^2 (4 Σ4i^2 + Ck (Σ3i - Σ2i) Σ1i) τi[t*]^2;
Tpver = Tpver + 2 U b Ck (Σ3i - Σ2i) f τi[t*] (h*)′[t*];
Tpver = Tpver + U b (2 Σ5ii + (Ck (2 ci[0] + ci[1]) - ci[1]) (Σ3i - Σ2i)) f τi[t*] τi′[t*];
Tpver = Tpver + b^2 ci[0] ci[1] f^2 τi[t*] τi″[t*];
Tpver = Tpver + b^2 ci[1] f^2 τi[t*] (h*)″[t*];
Tpver = -(π ρ b) Tpver;
Tp - Tpver // Simplify (* Should be zero *)

Out[70]= 0
```

- **Power summation verification**

```
In[71]:= Pver = U^2 (ci[1] Σ1i - Σ5ii + Ck (2 ci[0] - ci[1]) Σ1i ) f τi[t*] τi′[t*];
Pver = Pver + 2 U^2 Ck Σ1i f τi[t*] (h*)′[t*];
Pver = Pver + 1/2 U b (2 ci[0] + ci[1]) (Ck (2 ci[0] - ci[1]) + ci[1]) f^2 τi′[t*]^2;
Pver = Pver + 2 U b Ck f^2 (h*)′[t*]^2;
Pver = Pver + U b (4 Ck ci[0] + ci[1]) f^2 τi′[t*] (h*)′[t*];
Pver = Pver + b^2 (ci[0]^2 + 1/8 ci[1]^2 - ci[0] ci[2] + Σ6ii - Σ7ii) f^3 τi′[t*] τi″[t*];
Pver = Pver + 1/2 b^2 (2 ci[0] - ci[2]) f^3 τi″[t*] (h*)′[t*];
Pver = Pver + 1/2 b^2 (2 ci[0] - ci[2]) f^3 τi′[t*] (h*)″[t*];

Pver = Pver + b^2 f^3 (h*)′[t*] (h*)″[t*];
Pver = -(π ρ b^2) Pver;
P - Pver // Simplify (* Should be zero *)

Out[81]= 0
```

- **Kappa summation verification**

```
In[82]:= c = 2 b;
k = (π f c)/U; (*reduced frequency*)
κm = k^2/(4 π) (ci[0] cj[0] + 1/8 ci[1] cj[1] - 1/2 (ci[0] cj[2] + ci[2] cj[0]) + Σ6ji - Σ7ji);
κc = k/4 (Ck (2 ci[0] + ci[1]) (2 cj[0] - cj[1]) + ci[1] (2 cj[0] + cj[1]) + 2 Σ8ji);
κk = π (Ck (2 cj[0] - cj[1]) Σ1i + cj[1] Σ1i - Σ5ji);
μ = k^2/(8 π) (2 cj[0] - cj[2]);
η = k/2 Ck (2 cj[0] - cj[1]);
Qjver = -κm τi″[t*] - κc τi′[t*] - κk τi[t*] - μ (h*)″[t*] - η (h*)′[t*];
Qj* - Qjver // Simplify (* Should be zero *)

Out[90]= 0
```